Meetkunde met de Micro

H.A. Lauwerier
M.I., U.v.A., Amsterdam

Samenvatting

De microcomputer met grafisch beeldscherm kan voor het wiskunde-onderwijs van grote betekenis worden. Vooral het meetkunde-onderwijs kan er enorm van profiteren. Schrijver roept een beeld op van toekomstig, maar nu technisch realiseerbaar, meetkunde-onderwijs. Een aantal concrete toepassingen van stereometrisch tekenen m.b.v. de computer worden gegeven.

Summary

The micro-computer with a graphics-screen may prove to be of great importance for math-education. Especially geometry-education may benefit. The author tries to sketch the possibilities can be realised from a technical point of view. He focusses on applications of three-dimensional drawings with the computer.

......het wiskundepracticum was altijd een van de leuke uurtjes. Er waren tien microcomputers met elk een beeldscherm en een geheugeneenheid, maar elk apparaat was ook gekoppeld aan het grote beeldscherm dat de plaats innam waar vroeger het schoolbord was. De les ging over de omwentelingshyperboloïde. De leraar vertelde dat je ze in achtergebleven industriële gebieden nog kon tegenkomen als landschap ontsierende koeltorens en dat hun constructie gebaseerd was op rechte lijnen in de vorm van betonijzer als versterking van het beton.
De thuis voorbereide programma's kon je op je eigen diskette meenemen en aan de leraar ter controle voorleggen. De beste programma's en tekeningen werden op het grote scherm geprojecteerd en besproken. Voor de mooiste tekeningen werd de computeroutput naar de centrale plotter gedirigeerd. Natuurlijk waren er altijd wel een paar leerlingen die stiekem een geheime 'file' op het centrale geheugen hadden gereserveerd en via een 'store' en 'load' opdracht boodschapjes uitwisselden.

Science fiction of realiteit? Technisch is het al mogelijk. De ingrediënten zijn een lokaal met computerapparatuur, een overhead projector met een projectiescherm, een enthousiaste en deskundige leraar en leerlingen die hebben ontdekt dat spelen met wiskunde op de computer veel leuker kan zijn dan het steeds herhalen van dezelfde star wars spelletjes.

De ontwikkelingen van de techniek zijn dan ook verbijsterend. Het is nog maar luttele jaren geleden dat de (programmeerbare) zakcomputer zijn intrede deed in het onderwijs als een eendimensionale getallenproducerende rekenautomaat. Maar nu hebben we de beschikking over een *tweedimensionale* wiskunderobot die niet alleen kan rekenen maar die ook kan tekenen. De invloed op het wiskunde-onderwijs zal enorm zijn. Voor het eerst in de geschiedenis beschikt de wiskunde over een instrument, de microcomputer, waarmee geëxperimenteerd kan worden. In de frontlinie van de wetenschap zijn er al voorbeelden dat computer-experimenten konden bijdragen tot beter inzicht in de theorie en konden leiden tot de ontwikkeling van nieuwe theoretische resultaten. Voor de school zal de computer vooral een instrument zijn dat ons inzicht verschaft in wiskundige structuren, een instrument dat snel en doeltreffend rekent en dat het rekenwerk weet om te zetten in grafische voorstellingen, in punten, lijnen en krommen. Als mens, homo faber, zijn we toegerust met de mogelijkheid werktuigen, bouwsels, ruimtelijke constructies in elkaar te zetten. Daartoe is ruimtelijk inzicht nodig. Op de maatschappij gericht wiskunde-onderwijs dient de jonge mens dan ook al vroeg vertrouwd te maken met meetkundige begrippen en ruimtelijke constructies. De overheid heeft dit terecht begrepen door de meetkunde nu weer in het schoolprogramma op te nemen. Meetkunde is terug van weggeweest. Maar

wat niet mag terugkeren is de manier waarop de meetkunde vroeger behandeld werd. Meetkunde is altijd belangrijk geweest. De Oude Grieken en Romeinen zijn ons voorgegaan zowel in de theorie als in de praktijk, bijv. bij het boren van een tunnel of het aanleggen van een aquaduct. Veel later, de tijd van de ontdekkingsreizen, van Columbus en Mercator, kwam de meetkunde tot nieuwe bloei, daartoe gestimuleerd door de eisen van de cartografie en landmeetkunde. Ook militaire problemen deden hun invloed gelden en nieuwe vakken als beschrijvende meetkunde, een soort bouwkundig tekenen, kwamen tot ontwikkeling vanuit de vestingbouwkunde.
Tot voor kort is het wiskunde-onderwijs nog geheel in de ban geweest van de landmeters en de vestingbouwers. Enige decennia geleden bevatte de eerste middelbare akte van de a.s. wiskundeleraar, de akte K1, als verplicht onderdeel nog boldriehoeksmeting en beschrijvende meetkunde in orthogonale en centrale projectie!

Het toverwoord van deze tijd CAD/CAM, d.w.z. tekenen en ontwerpen m.b.v. de computer. De Teleac heeft het al in onze huiskamers gebracht. Velen die nu in de schoolbanken zitten zullen er later mee te maken hebben. Architecten, planologen, kunstenaars, grafisch ontwerpers, constructeurs, enz. enz.
Het principe van de moderne tekentechniek is de afbeelding van een ruimtelijk object op het beeldscherm van de computer in verschillende standen met verschillende projecties. Dat kunnen we op school doen met eenvoudige lichamen als een kubus of een kegel.
Wat kunnen wiskundedocenten nu daadwerkelijk doen om althans een eerste begin te maken met de computerisering van de meetkunde?
Het probleem is wel dat niet alle op scholen aanwezige computers even bruikbaar zijn voor grafische doeleinden. Er is helaas te weinig nagedacht over aan welke specificaties een goede schoolcomputer voor wiskunde-onderwijs dient te voldoen. Een computer die geschikt is voor informatica-onderwijs kan volkomen ongeschikt zijn voor het wiskunde-onderwijs. Wanneer een computer niet op eenvoudige wijze plaatjes kan tekenen hebben we er niets aan en kunnen we beter een programmeerbare zakcomputer gebruiken.
Een ander probleem is dat niet alle computers dezelfde programmeertaal gebruiken. Voor schoolgebruik is Basic een prima taal maar afhankelijk van het besturingssysteem heeft bijna elk type zijn eigen versie. Zeker in de, voor meetkunde noodzakelijke, grafische opdrachten zijn er grote onderlinge verschillen. Dit maakt het schrijven van alom bruikbare programma's tot een nagenoeg onmogelijke zaak.

Voor een wiskundige die gewend is gecompliceerde zaken te herleiden tot elementaire principes is dit niet zo erg. Het blijkt dat alle grafische opdrachten herleid kunnen worden tot maar drie elementaire opdrachten. We stellen nu de ideale schoolcomputer aan u voor die naast de gebruikelijke, algemeen beschaafde, Basic opdrachten de volgende grafische kent:

```
scale  a,b,c,d
move   p,q
draw   r,s
```

Het beeldscherm stellen we ons voor als een vierkant (schoolbord) met een plaatsbepaling in Cartesische coördinaten x,y.
Met scale a,b,c,d bedoelen we dat de linkerrand bepaald is door $x = a$ en de rechterrand door $x = b$, evenzo dat onderrand en bovenrand corresponderen met resp. $y = c$ en $y = d$.
Het tekenen geschiedt met een elektronisch krijtje of pen. De opdracht:

```
move  p,q
```

betekent dat de pen gebracht wordt naar het punt met de coördinaten (p,q). Er wordt dan nog niets getekend. Dat gebeurt met de opdracht:

```
draw  r,s
```

Bevindt de pen zich in de positie (p,q) dan wordt er nu een recht lijntje getrokken tot aan het punt (r,s). De pen blijft er dan staan, gereed om aan een volgend lijntje te beginnen.
Dat is alles. Er zijn computers waar deze drie opdrachten tot het standaardrepertoire behoren. Daarnaast zijn er vaak een aantal opdrachten die er alleen maar toe dienen om plaatjes op geriefelijke wijze van verfraaiingen te voorzien maar we behoeven er geen gebruik van te maken. Voordat we het schoolbord gebruiken, maken we het eerst schoon. Evenzo laten we computerprogramma's beginnen met de opdracht:

```
gclear
```

d.w.z. graphics clear, waarmee het beeldscherm is schoongemaakt.

Voorbeeld
Om een vierkant te tekenen kunnen we bijv. het volgende programma uitvoeren:

```
10   gclear
20   scale -2, 2, -2, 2
30   move 1, 0
40   draw 0, 1
50   draw -1, 0
60   draw 0, -1
70   draw 1, 0
80   end
```

Voorbeeld
Met het volgende programma kan een regelmatige n-hoek getekend worden. De coördinaten van de n hoekpunten $P_0, P_1 \ldots, P_{n-1}$ kiezen we als:

$$\begin{cases} x_k = \cos(k \cdot 360/n) \\ y_k = \sin(k \cdot 360/n) \end{cases}$$

met $k = 0, 1, \ldots, n$.
De opdracht deg geeft aan dat de berekening in graden is.

```
10   gclear
20   deg
30   scale -1, 1, -1, 1
40   disp "kies aantal zijden"
50   input n
60   move 1, 0
70   for k = 0 to n
80   draw cos (k * 360/n), sin (k * 360/n)
90   next k
100  end
```

De opdracht disp (display) is slechts een geheugensteuntje als voorbereiding voor de input waarbij de waarde van n via het toetsenbord wordt ingevoerd.

Wie een dergelijk programma op zijn computer uitvoert, zal bemerken dat de kwaliteit van de plaatjes nadelig beïnvloed wordt door het in het algemeen nogal matig oplossend vermogen van het beeldscherm. Het verschil tussen een regelmatige 30-hoek en een 60-hoek is dan ook nauwelijks te zien en het plaatje zouden we net zo goed kunnen zien als dat van een cirkel. Wanneer we dus een cirkel willen tekenen vervangen we de cirkel door een aaneenschakeling van veel kleine rechte lijntjes, net zoveel als in de gegeven situatie nodig is.

Wanneer we op het beeldscherm een kromme willen tekenen kunnen we het beste gebruikmaken van een parametervoorstelling als:

(1) $x = f(t),\ y = g(t)$

waarbij we t, bijv. vanaf 0, met kleine stapjes h laten aangroeien en dan telkens een klein lijntje tekenen. Het programma is bijv. in principe:

```
move  f(0), g(0)
for   t = 0 to n * h step h
draw  f(t), g(t)
next  t
```

of

```
t = 0
move  f(0), g(0)
t = t + h
draw  f(t), g(t)
if...   then...
goto
```

Aan een vergelijking als $x^2 + y^2 = 1$ hebben we niet veel, maar de parametervergelijking:

$x = \cos t,\ y = \sin t$

is veel bruikbaarder.
Deze constatering heeft consequenties voor onze kijk op functies en krommen. Het is blijkbaar nuttig enige vaardigheid te ontwikkelen in het hanteren van door parametervergelijkingen gegeven functies. Even een klein voorbeeldje.

Voorbeeld
Door de vergelijking:

$x^2 + xy + y^2 = 1$

is een functioneel verband vastgelegd tussen x en y, of, in meetkundige termen, wordt een kromme bepaald. Hoe analyseren we dit met de computer?
Door kwadraatafsplitsing, een veel gebruikte truc, maken we er van:

$\frac{3}{4}x^2 + (\frac{1}{2}x + y)^2 = 1,$

we denken aan:

$\cos^2 t + \sin^2 t = 1$

en stellen dus:

$x = \frac{2}{\sqrt{3}}\cos t,\ y = -\frac{1}{2}x + \sin t.$

Vervolgens voeren we het volgende programma uit:

```
10  gclear
20  scale -2, 2, -2, 2
30  move 1.1547, -0.5774
40  for t = 0 to 360 step 4
50  x = 1.1547 * cos (t)
60  y = sin (t) - x/2
70  draw x, y
80  next t
90  end
```

De kromme (fig. 1) ziet eruit als een ellips met symmetrieassen $x \pm y = 0$. Nu komen de vragen. Hoe kunnen we het e.e.a. verklaren op grond van de theorie? enz. enz.



fig. 1

Maar de belangrijkste toepassing van de microcomputer is toch het stereometrisch tekenen, het projecteren van ruimtelijke figuren. We beperken ons daarbij tot de parallelprojectie. De ruimtelijke figuur denken we geplaatst in een driedimensionaal assenstelsel waarvan het beeldscherm het x,y-vlak is en waarbij de positieve z-as loodrecht op het beeldscherm naar ons toe wijst. Projecteren komt neer op het werpen van slagschaduw op het x,y-vlak bij beschijning van evenwijdig licht waarvan de richting door de vector (p,q,1) gegeven is. Een lichtstraal door het punt P_0 (x_0, y_0, z_0) wordt beschreven door:

(2) $x = x_0 - pt,\ y = y_0 - qt,\ z = z_0 - t$

in parametervorm. Het x,y-vlak wordt getroffen voor de parameterwaarde $t = z_0$. De coördinaten van het trefpunt, de projectie, zijn dus:

(3) $\begin{cases} x = x_0 - pz_0 \\ y = y_0 - qz_0 \end{cases}$

Projectie op het x,y-vlak, het beeldscherm, van een daaraan evenwijdig lopend lijnstuk levert een lijnstuk van dezelfde richting en grootte op, maar een lijnstuk loodrecht op het projectievlak, bijv. de eenheidsvector (0,0,1) langs de z-as ondergaat een verkorting. Inderdaad volgt uit bovenstaande projectieregel (3) dat de ruimtelijke vector (0,0,1) overgaat in de tweedimensionale vector $(-p, -q)$. De verkorting is dus $\sqrt{p^2 + q^2}$. We kunnen het e.e.a. illustreren aan de hand van de bekende tekening van de doorzichtige kubus.



fig. 2

De kubusribbe OE, ruimtelijk corresponderend met de eenheidsvector (0,0,1) ziet er in projectie uit als een lijnstuk met de verkortingsverhouding $\frac{1}{2}$ onder een hoek van 30° met de horizontaal.



fig. 3

Er geldt dan:

$$q/p = \tan\alpha = \frac{1}{\sqrt{3}}$$

en

$$\sqrt{p^2 + q^2} = \frac{1}{2}$$

De projectieparameters zijn dus:

$$p = 0.433, \quad q = 0.250$$

De formules (3) stellen ons in staat om op eenvoudige en realistische wijze tekeningen, projecties dus, van ruimtelijke figuren op beeldschermen of tekenvlak te vervaardigen. We geven er twee toepassingen van, de schroeflijn of kurketrekkerkromme en de omwentelingshyperboloïde waarmee we dit verhaal begonnen zijn.

De kurketrekker combineert een circulaire draaiing:

$$y = \cos t, \quad z = \sin t$$

met een voorwaartse beweging in de richting van de positieve x-as:

$$x = at, \quad a > 0.$$

Met nog vrij te kiezen projectiegetallen p,q geeft dan formule (3) de parametervoorstelling van de, op het beeldscherm te tekenen, kromme:

$$\begin{cases} x = at - p\sin t, \\ y = \cos t - q\sin t. \end{cases}$$

Met bijv. het volgende programma:

```
10 REM ***SCHROEFLIJN***
20 GCLEAR
30 RAD
40 DISP "VOER PARAMETER A IN"
50 INPUT A
60 DISP "VOER PROJECTIEPARAMETERS P,Q IN"
70 INPUT P,Q
80 SCALE -P,20*A*PI+P,-2,2
90 MOVE 0,1
100 FOR T=0 TO 20*PI STEP PI/40
110 X=A*T-P*SIN(T)
120 Y=COS(T)-Q*SIN(T)
130 DRAW X,Y
140 NEXT T
150 END
```

kan een plaatje als fig. 4 gemaakt worden. Het zou een aardige opgave voor een wiskundepracticum zijn de leerlingen zelf te laten experimenteren met verschillende waarden voor a, p en q en de invloed hiervan op het resultaat te onderzoeken.



fig. 4

Wanneer een rechte lijn gewenteld wordt om een draaiingsas in kruisende positie ontstaat een z.g. omwentelingshyperboloïde, de koeltorens van de Staatsmijnen bijv. Van dit eenvoudige meetkundige feit is niet iedereen op de hoogte. Het valt niet mee het zich ruimtelijk voor te stellen en om, op de ouderwetse wijze, er een overtuigende tekening van te maken, maar met de computer gaat het weer eenvoudig.

In ruimtelijke Cartesische coördinaten kan een omwentelingshyperboloïde voorgesteld worden door bijv. de vergelijking:

$$(4) \qquad \frac{x^2 + z^2}{a^2} - \frac{y^2}{b^2} = 1.$$

Het snijvlak $z = 0$ levert een hyperbool en elk snijvlak y = constant geeft een cirkel als doorsnede. De hyperboloïde is dus rotatie-symmetrisch met de y-as als rotatieas. We kunnen gemakkelijk laten zien dat er op het oppervlak van de hyperboloïde rechte lijnen aanwezig zijn. Daartoe gebruiken we een eenvoudig criterium. Normaal zal een rechte lijn een hyperboloïde, en trouwens elk ander tweedegraadsoppervlak, in hoogstens twee punten kunnen snijden. In een uitzonderlijke situatie waarbij er meer, minstens drie, snijpunten zijn moet de lijn deel uitmaken van het oppervlak. Bij een kegel kan men dit meteen zien, maar bij een hyperboloïde is het niet anders. Welnu, de twee punten $P(ac/b,c,a)$ en $Q(-ac/b, -c,a)$ liggen blijkbaar beide op de door (4) bepaalde hyperboloïde. Het midden van PQ met de coördinaten $(0,0,a)$ ligt er eveneens op en daarmee de hele lijn.



fig. 5

In het computerprogramma laten we P om de y-as over een cirkel draaien:

$$x = \cos t,\ y = 1,\ z = \sin t.$$

Op analoge wijze laten we Q over eenzelfde lagere cirkel, op het niveau $y = -1$, draaien maar daarbij laten we Q op P voorlopen:

$$x = \cos (t + \alpha),\ y = -1,\ z = \sin (t + \alpha)$$

In het programma kan de z.g. verdraaiingsparameter willekeurig gekozen worden. Met eveneens willekeurig te kiezen projectieparameter kan dan m.b.v. de formules (3) alles getekend worden. Overigens hebben we daarbij $p = 0$ genomen. In de regels 130-160 wordt de cirkelvormige bovenbegrenzing van de hyperboloïde, d.w.z. de baan van P getekend, een ellips als projectie van een cirkel. In de regels 170-200 gebeurt hetzelfde aan de onderkant. In de regels 210-240 gebeurt het eigenlijke werk, het tekenen van de projecties van een vooraf te kiezen aantal posities van de wentelende lijn PQ.

```
10 REM "HYPERBOLOIDE"
20 SCLEAR
30 RAD
40 DISP "KIES PROJECTIEPARAMETER"
50 DISP "NEEM BIJV.Q=0.3"
60 INPUT Q
70 SCALE -1-Q,1+Q,-1-Q,1+Q
80 DISP "KIES VERDRAAIINGSPARAMETER"
90 DISP "NEEM BIJV.A=2"
100 INPUT A
110 DISP "KIES AANTAL LIJNEN"
120 INPUT N
130 MOVE 1,1
140 FOR T=0 TO 2*PI STEP PI/50
150 DRAW COS(T),1-Q*SIN(T)
160 NEXT T
170 MOVE 1,-1
180 FOR T=0 TO 2*PI STEP PI/50
190 DRAW COS(T),-1-Q*SIN(T)
200 NEXT T
210 FOR T=0 TO 2*PI STEP 2*PI/N
220 MOVE COS(T),1-Q*SIN(T)
230 DRAW COS(T+A),-1-Q*SIN(T+A)
240 NEXT T
250 END
```

Het resultaat is fig. 6. Ook hier weer kan naar hartelust geëxperimenteerd worden, bijv. door verschillende waarden van de verdraaiingshoek α te kiezen. Het is daarbij voldoende $0 \leq \alpha \leq \pi$ te kiezen. De uiterste waarden leveren iets speciaals op. Voor $\alpha = 0$ komt er een cylinder en voor $\alpha = \pi$ een kegel. Uiteraard kan het een en ander met wat coördinatenmeetkunde bevestigd worden. De gevormde hyperboloïde heeft de vergelijking.

$$2(x^2 + z^2) - (1 - \cos \alpha)y^2 = 1 + \cos \alpha,$$

en ook daaraan kan men o.a. zien wat er gebeurt voor bijv. $\alpha = 0$ en $\alpha = \pi$.



fig. 6